# $Id$
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: François Charette <firmicus ατ gmx δοτ net>

pkgname=perl-mime-tools
pkgver=5.509
pkgrel=3
pkgdesc="Parses streams to create MIME entities"
arch=('any')
url="http://search.cpan.org/dist/MIME-tools"
license=('GPL' 'PerlArtistic')
depends=('perl-mailtools' 'perl-io-stringy' 'perl-convert-binhex')
options=('!emptydirs')
provides=('perl-mime-body=5.428.0' 'perl-mime-decoder' 'perl-mime-decoder-base64'
	'perl-mime-decoder-binary' 'perl-mime-decoder-binhex' 'perl-mime-decoder-gzip64'
	'perl-mime-decoder-nbit' 'perl-mime-decoder-quotedprint' 'perl-mime-decoder-uu'
	'perl-mime-entity' 'perl-mime-field-contdisp=5.428.0' 'perl-mime-field-contraenc=5.428.0'
	'perl-mime-field-conttype=5.428.0' 'perl-mime-field-paramval=5.428.0' 'perl-mime-head=5.428.0'
	'perl-mime-parser' 'perl-mime-parser-filer' 'perl-mime-parser-reader'
	'perl-mime-parser-results' 'perl-mime-worddecoder' 'perl-mime-words=5.428.0')
source=("http://www.cpan.org/authors/id/D/DS/DSKOLL/MIME-tools-${pkgver}.tar.gz") 
md5sums=('a2f0e07da2ff4b0478908544cc4b40fa')

build() {
  cd  "${srcdir}/MIME-tools-${pkgver}"

  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd  "${srcdir}/MIME-tools-${pkgver}"

  make install DESTDIR="${pkgdir}"
}
